

# 【AM・FM変調/復調 AM・FM Modulation/Demodulation】

## ■ AM変調

$s(t)$ :変調信号, $A$ :搬送波振幅, $m$ :変調度とする.

$$\text{AM波: } g(t) = (1 + ms(t))A\sin(\omega_0 t)$$

## ■ AM復調

AM波の解析信号を  $\overline{g(t)}$  とすると,

$$\overline{g(t)} = (1 + ms(t))(\sin(\omega_0 t) + j\cos(\omega_0 t)) = (1 + ms(t)) \exp\{j(\frac{\pi}{2} - \omega_0 t)\}$$

但し, $A = 1$ とした.  $1 + ms(t) \geq 0$ を仮定し,絶対値をとると,

$$|\overline{g(t)}| = |(1 + ms(t))| \cdot |\exp\{j(\frac{\pi}{2} - \omega_0 t)\}| = 1 + ms(t)$$

以上より,復調信号を得る.

## ■ FM変調・復調




Figure 1: FM変調ブロック図

$C0$ :搬送波の周波数を決める定数  $C1$ :入力信号の振幅に対する周波数の変化の割合を決める定数




Figure 2: FM復調ブロック図





Figure 3: Scilab 実行結果

Source Code 1: Scilab

```

////////////////////////////////////
//          AM/FM変復調
//  AM FM modulation/demodulation
//
//                               M.Tsutsui
////////////////////////////////////

clear;

d_size=200;

t=linspace(0,100,d_size);

//_____変調信号_AMM/FM共通_____//

f=2000;//変調信号 周波数
omega=2*%pi*f;
s=sin(omega*t);//変調信号

//-----AM変復調-----//

//__キャリア 設定__//

fc=10000;//キャリア周波数

A=1;//振幅

omega_c=2*%pi*fc;

carrier=A*sin(omega_c*t);

//-----//

```

```

m=0.8;//変調度

AM_wave=(1+m*s)*A.*carrier;//AM波

//_____AM復調処理_____//

AM_H=hilbert(AM_wave);
AM_dem=abs(AM_H);

subplot(3,2,1)
plot(t,s);
xgrid();
title('Modulating_Signal(AM)', 'fontsize',4);
subplot(3,2,3)
plot(t,AM_wave, 'r');
xgrid();
title('Modulated_Signal(AM)', 'fontsize',4);
subplot(3,2,5)
plot(t,AM_dem, 'k');
xgrid();
title('AM_demodulation', 'fontsize',4);

//_____FM変復調_____//

//_____FM変調処理_____//

c1=0.5;
c0=0.83;

phi(1)=0;

for n=2:1:d_size;
    phi(n)=phi(n-1)+c0+c1*s(n-1);
end
yn=sin(phi)';//FM波

//_____FM復調処理_____//

as=hilbert(yn);//解析信号化
arg=atan(imag(as)./(real(as)+%eps));//偏角

for i=2:1:d_size;
    FM_demod(i-1)=arg(i)-arg(i-1);//復調信号
end

subplot(3,2,2);
plot(s);
xgrid();
title('Modulating_Signal(FM)', 'fontsize',4);
subplot(3,2,4);
plot(yn, 'r');
xgrid();
title('Modulated_Signal(FM)', 'fontsize',4);
subplot(3,2,6);
plot(FM_demod, 'k');
a=gca();
a.data_bounds(:,2)=[0;1.5];
xgrid();
title('FM_demodulation', 'fontsize',4);

```